

PLOT**PURPOSE**

Generates a plot.

DESCRIPTION

The PLOT command allows the analyst to generate single or multi-trace plots of data, functions, or both. It is DATAPLOT's most powerful, most important, and most heavily used graphics command. There are 7 general plot syntaxes:

1. 1-variable form
2. 2-variable form
3. 3-variable multi-trace form
4. VERSUS form
5. multi-VERSUS form
6. function form
7. AND form

DATAPLOT uses the concept of traces. A trace is a connected set of points. Points in the same trace are plotted with the same attributes. In most cases, a single variable is one trace. However, a single variable can be split into multiple traces if desired (see SYNTAX 3).

SYNTAX 1 (1-variable form)

PLOT <y> <SUBSET/EXCEPT/FOR qualification>

where <y> is a variable;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This form for the PLOT command is used for plotting <y> versus its dummy index. The resulting plot will have <y> on the vertical axis and the dummy index 1, 2, 3, ..., n (where n = the number of elements in <y>) on the horizontal axis. Some examples are:

```
PLOT Y
PLOT TEMP SUBSET TAG > 4
```

SYNTAX 2 (2-variable form)

PLOT <y> <x> <SUBSET/EXCEPT/FOR qualification>

where <y> is the dependent (i.e., the vertical axis) variable;

<x> is the independent (i.e., the horizontal axis) variable;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This is the 2-argument form for the PLOT command. It is used for plotting <y> versus <x>. The resulting plot will have <y> on the vertical axis and <x> on the horizontal axis. Some examples are:

```
PLOT Y X
PLOT RES X SUBSET X > -9999
```

SYNTAX 3 (the 3-variable multi-trace form)

PLOT <y> <x> <tag> <SUBSET/EXCEPT/FOR qualification>

where <y> is the dependent (i.e., the vertical axis) variable;

<x> is the independent (i.e., the horizontal axis) variable;

<tag> is a variable that identifies groups in <y> and <x> that are plotted with common attributes;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This is the 3-argument form for the PLOT command. It is used for multi-trace plotting of <y> versus <x>. The resulting plot will have <y> on the vertical axis, <x> on the horizontal axis, and will have one trace for each distinct value in the <tag> variable. Some examples are:

```
PLOT Y X LAB
PLOT PRES TEMP DAY
PLOT PRES TEMP DAY SUBSET DAY <> 4
```

If the <x> variable and the <tag> variable are identical, all points with a common <x> value are treated as a common trace (i.e., they are plotted with common attributes).

Although DATAPLOT supports a large number of built-in plot formats, there will be cases where you may want a specialized chart format that is not available. This syntax for the PLOT command can often be used for this purpose by defining the <tag> variable in the right way. Points with a common <tag> value are treated as a trace, and attributes can be set for each individual trace.

SYNTAX 4 (VERSUS form)

PLOT <y1> <y2> <y3> ... <yk> VERSUS <x> <SUBSET/EXCEPT/FOR qualification>

where <y1>, <y2>, <y3>, ..., <yk> are dependent (i.e., vertical axis) variables;

<x> is an independent (i.e., the horizontal axis) variable;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This is the single-VERSUS argument form for the PLOT command. It is used for multi-trace plotting where the dependent variables are plotted against a common <x> variable. The resulting plot will have one trace for each <yi> variable:

<y1> (vertically) versus <x> (horizontally)

<y2> (vertically) versus <x> (horizontally)

<y3> (vertically) versus <x> (horizontally)

...

<yk> (vertically) versus <x> (horizontally)

Some examples are:

```
PLOT Y1 Y2 Y3 VERSUS X
```

```
PLOT Y PRED VERSUS X
```

```
PLOT Y PRED VERSUS X SUBSET X = 10.6 TO 19.7
```

SYNTAX 5 (multi-VERSUS form)

PLOT <syntax 4> <syntax 4> ... <syntax 4>

This is the multi-VERSUS argument form for the PLOT command. It is used for multi-trace plotting where the dependent variables are plotted against different <x> variables. Some examples are:

```
PLOT Y1 Y2 Y3 VERSUS X1 Y4 Y5 VERSUS X2
```

```
PLOT P1 VERSUS T1 P2 VERSUS T2 P3 VERUS T3
```

SYNTAX 6 (function form)

PLOT <f> FOR <x> = <start> <increment> <stop>

where <f> is a function (either pre-defined via the LET FUNCTION command, or explicitly defined herein);

<x> is the dummy variable in the function;

<start> is the desired minimum value for <x> at which the function is to be evaluated;

<increment> is the desired increment value for <x> at which the function is to be evaluated;

and <stop> is the desired maximum value for <x> at which the function is to be evaluated.

This is the function form for the PLOT command. It is used for plotting a function of one variable. Some examples are:

```
PLOT SIN(X)*EXP(-X) FOR X = 0 .1 5
```

```
LET FUNCTION F = EXP(-X*SIN(X**2))
```

```
PLOT F FOR X = 0 .1 3
```

SYNTAX 7 (AND form)

<any valid syntax 1 to 6> AND

<any valid syntax 1 to 6> AND

<any valid syntax 1 to 6> AND

...

<any valid syntax 1 to 6> AND

<any valid syntax 1 to 6>

This is the most general syntax for PLOT. It is used for generating multi-trace plots of variables, of functions, or of mixtures of both.

Some examples are:

```
PLOT Y X AND
```

```
PLOT A+B*X FOR X = 1 1 10
```

```
PLOT Y1 Y2 VS X AND
PLOT Y X AND
PLOT A*SIN(B*X) FOR X = 1 .1 3 AND
PLOT Y3 X3 LAB
```

NOTE 1

Plot points can be plotted as characters, connected lines, spikes, or bars. These are set independently of each other. The default is to plot each trace as a connected line with no symbol, no bar, and no spike. The **LINE**, **CHARACTER**, **SPIKE**, and **BAR** commands are used to set the switches for plotting a given trace as a connected line, a character, a spike, or a bar respectively.

There are attribute setting commands for lines, characters, spikes, and bars. See the documentation for **LINE**, **CHARACTER**, **SPIKE**, and **BAR** for a complete list of these commands. Attributes are set giving a list of values. The first trace uses the first setting, the second trace uses the second setting, and so on. For example, **CHARACTER SIZE 2.0 3.0 1.5** sets the character size for trace 1 to 2.0, the character size for trace 2 to 3.0, and the character size for trace 3 to 1.5. Attributes can be set for up to 100 traces.

As a more complex example, suppose you want to plot a variable **Y** as a connected line and every fifth point as a filled circle. You can do something like the following:

```
LET N = SIZE Y
LET X = SEQUENCE 1 1 N
LET TAG = PATTERN 1 2 2 2 2 FOR I = 1 1 N
CHARACTER CIRCLE BLANK
CHARACTER FILL ON OFF
CHARACTER SIZE 1.5
PLOT Y X TAG
```

NOTE 2

DATAPLOT provides a large range of plot control features for the plot. This includes titles, axis labels, legends, and so on. **DATAPLOT** sets these with separate commands (as opposed to arguments on the **PLOT** command itself). Each of these features typically has its own attribute setting commands as well. **DATAPLOT** simply uses whatever the current setting is for each of these attributes when it generates a plot. For example, a **TITLE** command is entered to define the plot title (nothing is actually generated until the next **PLOT** is performed). This title remains in effect for all subsequent plots until it is changed (another **TITLE** command) or deleted (**TITLE** with no arguments).

Most of the commonly used plot features are listed below in the **RELATED COMMANDS** section. The attribute setting commands are not listed (e.g., **TITLE** is listed, but **TITLE COLOR** and **TITLE SIZE** are not). See the documentation for the plot feature command for its attribute setting commands. These attribute setting commands are documented in the **Plot Control** chapter.

DEFAULT

None

SYNONYMS

VS and **VS.** are synonyms for **VERSUS**.

RELATED COMMANDS

CHARACTERS	=	Sets the types for plot characters.
LINES	=	Sets the types for plot lines.
SPIKES	=	Sets the on/off switches for plot spikes.
BARS	=	Sets the on/off switches for plot bars.
TITLE	=	Sets the plot title.
LABEL	=	Sets the plot axis labels.
LEGEND	=	Sets the plot legends.
BOX COORDINATES	=	Sets the locations for plot boxes.
ARROW COORDINATES	=	Sets the locations for plot arrows.
SEGMENT COORDINATES	=	Sets the locations for plot segments.
FRAME	=	Sets the on/off switch for the plot frame.
FRAME COORDINATES	=	Sets the location for the plot frame.
GRID	=	Sets the on/off switch for the plot grid.
LOG	=	Sets the on/off switch for log scale.
TIC	=	Sets the on/off switch for the plot tics.
TIC LABEL	=	Sets the on/off switch for the plot tic labels.

MARGIN COLOR	=	Sets the color for the plot margin.
BACKGROUND COLOR	=	Sets the color for the plot background.
PRE-ERASE	=	Sets the automatic pre-erase switch for plots.
SEQUENCE	=	Sets the automatic sequence switch for plots.
MULTILOT	=	Generate multiple plots per page.

APPLICATIONS

Data Analysis, Presentation Graphics

IMPLEMENTATION DATE


Pre-1987

PROGRAM 1

```

. THIS SAMPLE PROGRAM READS THE FILE BOXJE142.DAT IN THE DATAPLOT
. REFERENCE DIRECTORY. THESE DATA ARE YIELD FROM AN INDUSTRIAL PROCESS.
.
SKIP 25
READ BOXJE142.DAT YIELD
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE TIME SERIES PLOT
YILABEL YIELD
XILABEL SEQUENCE NUMBER
XLIMITS 0 70
XTIC OFFSET 2 2
PLOT YIELD
.
LINE BLANK
CHARACTER CIRCLE; CHARACTER SIZE 1.2
PLOT YIELD
.
LET N = SIZE YIELD; LET X = DATA 1 N
LET A = MEAN YIELD; LET Y = DATA A A
CHARACTER OFF
SPIKE ON; SPIKE BASE A
PLOT YIELD AND
PLOT Y X
.
SPIKE OFF; BAR ON
BAR BASE A; BAR WIDTH 0.6; BAR FILL ON
PLOT YIELD AND
PLOT Y X
END OF MULTIPLY


```



PROGRAM 2

```
. POLLUTION SOURCE ANALYSIS, LLOYD CURRIE, DATE--1990
. SUBSET OF CURRIE.DAT REFERENCE FILE
LET ID2 = DATA 1 1 1 1 1 2 2 2 2 2 2 2 2 2 2 2
SERIAL READ LEAD
164 426 59 98 312 263 607 497 213 54 160 262 547 325 419 94 70
END OF DATA
SERIAL READ POT
106 175 61 79 94 121 424 328 107 218 140 179 246 231 245 339 99
END OF DATA

.
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE SCATTER PLOT; X1LABEL LEAD; Y1LABEL POTASSIUM
LINE BLANK ALL; CHARACTER CIRCLE; CHARACTER FILL ON
PLOT POT LEAD
CHARACTER CIRCLE SQUARE; CHARACTER FILL OFF ALL
TITLE SCATTER PLOT WITH GROUPS
LEGEND 1 CIRC() - GROUP 1; LEGEND 2 SQUA() - GROUP 2
LEGEND FILL ON; LEGEND FONT DUPLEX
PLOT POT LEAD ID2
CHARACTER CIRCLE CIRCLE SQUARE SQUARE; CHARACTER FILL OFF ON OFF ON
LET X = SEQUENCE 1 1 17; LEGEND 1 CIRC() - POTASSIUM; LEGEND 2 SQUA() - LEAD
X1LABEL SEQUENCE; Y1LABEL; TITLE CHARACTER FILL REPRESENTS GROUP ID
PLOT POT X ID2 AND
PLOT LEAD X ID2
CHARACTER BLANK ALL; LINE SOLID DASHSEGMENT 2 PATTERN DASH
SEGMENT 1 COORDINATES 16 85 19 85; SEGMENT 2 COORDINATES 16 81 19 81
TITLE MULTIPLE TRACES AS LINES
PLOT POT LEAD VS X
END OF MULTIPLY
```



PROGRAM 3

```

MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
LET FUNCTION F = (1/SQRT(2*PI))*EXP(-0.5*X**2)
LET FUNCTION D1 = DERIVATIVE F WRT X
LINES SOLID DOT DOT
TITLE PLOT A FUNCTION AND THE DERIVATIVE
PLOT F FOR X = -3 .1 3 AND
PLOT D1 FOR X = -3 .1 3
.
PRE-SORT OFF; FRAME OFF; DEGREES
LET THETA = SEQUENCE 0 10 1000; LET R = 2*THETA
LET Y = R*SIN(THETA); LET X = R*COS(THETA)
TITLE A POLAR COORDINATE FUNCTION
PLOT Y X
PRE-SORT ON; FRAME ON; DELETE Y X
.
SKIP 25
READ UGIANSKY.DAT Y1 Y2 LAB
LEGEND 1 INTERLAB ANALYSIS; TITLE YOUDEN PLOT
LINES BLANK ALL; CHARACTERS 1 2 3 4 5 6 7 8 9; CHARACTER SIZE 4 ALL
LIMITS 0 5.5
PLOT Y1 Y2 LAB
LEGEND 1; LIMITS
.
SKIP 25
READ CHWIRUT1.DAT Y X LAB
CHARACTER X ALL; CHARACTER SIZE 1.5 ALL; LINE SOLID ALL
TITLE SHOW SPREAD DUE TO REPLICATION
PLOT Y X X
END OF MULTIPLY

```

